


# Modellierung der Bewegung einer „Schneewalze“



Dr.rer.nat. Dr.techn. Martin Tutz

November 2013

# Inhaltsverzeichnis

<b>1</b>	<b>Einleitung</b>	<b>2</b>
<b>2</b>	<b>Zylinder auf der schiefen Ebene</b>	<b>2</b>
2.1	Herleitung der Bewegungsgleichung . . . . .	2
2.2	Lösung der Bewegungsgleichung . . . . .	3
<b>3</b>	<b>Die Schneewalze</b>	<b>4</b>
3.1	Herleitung der Bewegungsgleichung(en) . . . . .	4
3.2	Lösung der Bewegungsgleichung . . . . .	6
3.3	Beispiel 1 . . . . .	7
3.4	Beispiel 2 . . . . .	9
<b>A</b>	<b>Anhang – Quellcode des Maxima-Programms</b>	<b>11</b>

# 1 Einleitung


Der vorliegende Artikel behandelt die Modellierung einer „Schneewalze“. Diese entspricht einem auf einer schiefen Ebene abrollenden Zylinder dessen Radius (und damit verbunden auch dessen Masse bzw. Trägheitsmoment) in Abhängigkeit des zurückgelegten Weges zunimmt.

Zum besseren Verständnis der folgenden Ausführungen wird zuerst der Standardfall des Zylinders mit konstantem Radius auf der schiefen Ebene kurz zusammengefasst wiederholt.

## 2 Zylinder auf der schiefen Ebene

### 2.1 Herleitung der Bewegungsgleichung

Im Folgenden wird ausgehend von dieser Abbildung die Bewegungsgleichung für den auf der schiefen Ebene abrollenden Zylinder mit konstantem Radius  $r$  hergeleitet. Die Luftwiderstandskraft  $F_L$  sowie die Rollreibungskraft  $F_R$  bleiben dabei unberücksichtigt.



In der Abbildung bezeichnet  $F_G$  die im Schwerpunkt  $S$  angreifende Gewichtskraft.  $F_{GP}$  und  $F_{GN}$  sind die zugehörigen Komponenten parallel bzw. normal zur schiefen Ebene. Bei gegebenem Neigungswinkel  $\alpha$  gilt

$$F_G = mg, \quad (1)$$

$$F_{GP} = mg \sin(\alpha), \quad (2)$$

$$F_{GN} = mg \cos(\alpha), \quad (3)$$

wobei  $m$  die Masse und  $g$  die Erdbeschleunigung des Zylinders bezeichnen. Die Bewegungsgleichung (in  $x$ -Richtung) lautet somit<sup>1</sup>

$$m\ddot{x} = F_{GP} - F_M. \quad (4)$$

---

<sup>1</sup>Es gilt:  $(\dot{\phantom{x}}) = \frac{d(\phantom{x})}{dt}$  bzw.  $(\ddot{\phantom{x}}) = \frac{d^2(\phantom{x})}{dt^2}$ .

$F_M$  bezeichnet darin jene Kraft, die durch das Drehmoment  $M$  aufgrund der Drehung des Zylinders um seine Symmetrieachse durch das Abrollen auf der schiefen Ebene zustande kommt.  $x = x(t)$  steht für den zurückgelegten Weg in  $x$ -Richtung,  $\ddot{x}$  für die Beschleunigung.

In Analogie zum 2. Newtonschen Gesetz gilt bei Rotation

$$M = I\ddot{\varphi} = F_M r, \quad (5)$$

wobei  $\ddot{\varphi}$  die zum Drehwinkel  $\varphi = \varphi(t)$  gehörende Winkelbeschleunigung symbolisiert. Der zurückgelegte Weg  $x$  hängt mit dem Drehwinkel  $\varphi$  über den (konstanten) Radius  $r$  zusammen

$$x = r\varphi. \quad (6)$$

Differenziert man die letzte Beziehung zweimal, erhält man

$$\ddot{x} = r\ddot{\varphi}. \quad (7)$$

Drückt man daraus die Winkelbeschleunigung  $\ddot{\varphi} = \frac{\ddot{x}}{r}$  aus und setzt man diese in (5) ein, erhält man nach Division durch  $r$

$$F_M = I \frac{\ddot{x}}{r^2}. \quad (8)$$

Setzt man diese Beziehung sowie (2) in (4) ein, erhält man

$$m\ddot{x} = mg \sin(\alpha) - I \frac{\ddot{x}}{r^2}. \quad (9)$$

Durch Einsetzen des (bekannten) Trägheitsmoments des Zylinders  $I = \frac{1}{2}mr^2$  und Umstellen bzw. Vereinfachen gelangt man zur gesuchten Bewegungsgleichung

$$\ddot{x} = \frac{2}{3} g \sin(\alpha). \quad (10)$$

## 2.2 Lösung der Bewegungsgleichung

Die Lösung von (10) gestaltet sich aufgrund der Tatsache, dass die Beschleunigung  $\ddot{x}$  konstant ist, sehr einfach. Durch Integration erhält man die Geschwindigkeit

$$\dot{x} = \frac{2}{3} gt \sin(\alpha) + v_0, \quad (11)$$

wobei  $v_0$  die Integrationskonstante ist. Setzt man  $t = 0$ , erkennt man, dass  $v_0$  die Geschwindigkeit  $\dot{x}(t = 0)$  ist. Analog erhält man für den Weg

$$x = \frac{g}{3} t^2 \sin(\alpha) + v_0 t + x_0 \quad (12)$$

mit  $x_0 = x(t = 0)$ .

Zur Veranschaulichung der Lösung sei auf Abschnitt 3.4 verwiesen.

### 3 Die Schneewalze

Die grundsätzliche Vorgangsweise für den Fall mit veränderlichem Radius  $r$  ist ähnlich der im vorangegangenen Abschnitt vorgestellten, die Umsetzung jedoch ungleich aufwändiger. Es folgt eine Herleitung der Bewegungsgleichungen inklusive Berücksichtigung der Rollreibungskraft  $F_R$  und der Luftwiderstandskraft  $F_L$ .

#### 3.1 Herleitung der Bewegungsgleichung(en)

Es wird angenommen, dass der Radius  $r$  linear mit dem zurückgelegten Weg  $x = x(t)$  ansteigt, d.h.  $r = r(x(t))$  oder indirekt bzw. vereinfacht  $r = r(t)$ .<sup>2</sup> Mit der Konstante  $C$ , welche den Zuwachs pro Längeneinheit angibt, kann dies wie folgt formuliert werden

$$r = r_0 + Cx. \quad (13)$$

$r_0$  bezeichnet hierin den Radius des Zylinders zum Zeitpunkt  $t = 0$ . Die in der Folge benötigten Ableitungen des Radius lauten

$$\dot{r} = C\dot{x}, \quad (14)$$

$$\ddot{r} = C\ddot{x}. \quad (15)$$

Der Zusammenhang zwischen dem Drehwinkel  $\varphi = \varphi(t)$  und dem zurückgelegten Weg  $x$  lautet

$$x = r\varphi \Rightarrow \varphi = \frac{x}{r}. \quad (16)$$

Aufgrund der Tatsache, dass nun alle drei Größen von  $t$  abhängen, gilt (mit Hilfe der Produktregel) für die beiden Ableitungen

$$\dot{x} = \dot{r}\varphi + r\dot{\varphi} \Rightarrow \dot{\varphi} = \frac{\dot{x} - \dot{r}\varphi}{r}, \quad (17)$$

$$\ddot{x} = \ddot{r}\varphi + 2\dot{r}\dot{\varphi} + r\ddot{\varphi} \Rightarrow \ddot{\varphi} = \frac{\ddot{x} - \ddot{r}\varphi - 2\dot{r}\dot{\varphi}}{r}, \quad (18)$$

wobei die vorangegangenen Zusammenhänge  $r$  und  $\dot{r}$  bzw.  $\varphi$  und  $\dot{\varphi}$  zur Auswertung sukzessive in nachfolgende Formeln einzusetzen sind.

Durch die zeitliche Änderung des Radius  $r$  ändern sich auch die Masse  $m = m(t)$  sowie das Trägheitsmoment  $I = I(t)$  des Zylinders. Unter der Annahme, dass die Dichte  $\rho$  des Zylinders konstant ist, gelten

$$m = \rho \pi h r^2 \quad (19)$$

und

$$I = \frac{1}{2} m r^2 = \frac{1}{2} \rho \pi h r^4, \quad (20)$$

---

<sup>2</sup>In weiterer Folge wird für den Drehwinkel  $\varphi$ , die Masse  $m$ , das Trägheitsmoment  $I$ , den Impuls  $p_x$  und den Drehimpuls  $L$  sowie deren Ableitungen diese vereinfachte Schreibweise verwendet.

wobei  $h$  die Höhe (des liegenden) Zylinders (= Breite der Schneewalze) angibt. Die im Folgenden benötigten ersten Ableitungen vom  $m$  und  $I$  lauten

$$\dot{m} = 2\rho\pi hr\dot{r}, \quad (21)$$

$$\dot{I} = 2\rho\pi hr^3\dot{r}. \quad (22)$$

Aufgrund der zeitlichen Änderung der Masse  $m$  verallgemeinert sich das 2. Newtonsche Gesetz als Ableitung des Impulses  $p_x = p_x(t)$  zu (analog für die  $y$ -Richtung)

$$F_x = \frac{dp_x}{dt} = \frac{d}{dt}(m\dot{x}) = \dot{m}\dot{x} + m\ddot{x}. \quad (23)$$

Die Bewegungsgleichungen mit Berücksichtigung der Rollreibungskraft  $F_R$  und Luftwiderstandskraft  $F_L$  lauten (siehe obige Abbildung)

$$\dot{m}\dot{x} + m\ddot{x} = F_{GP} - F_M - F_R - F_L, \quad (24)$$

$$\dot{m}\dot{y} + m\ddot{y} = -F_{GN} + F_N, \quad (25)$$

wobei für  $F_{GP}$  und  $F_{GN}$  die Beziehungen (2) und (3) gelten und  $F_N$  die Normalkraft der Ebene (des Bodens) bezeichnet. Die Rollreibungskraft kann wie folgt angeschrieben werden

$$F_R = \mu F_N, \quad (26)$$

wobei  $\mu$  den Reibungskoeffizienten bezeichnet und  $F_N$  aus (25) bestimmt werden kann

$$F_N = \dot{m}\dot{y} + m\ddot{y} + F_{GN}. \quad (27)$$

Für die Luftwiderstandskraft  $F_L$  gilt näherungsweise<sup>3</sup>

$$F_L = \frac{1}{2}\rho_L c_w A \dot{x}^2, \quad (28)$$

wobei für die Schattenfläche  $A = 2rh$  einzusetzen ist.  $\rho_L$  bezeichnet die Dichte der Luft,  $c_w$  den Widerstandsbeiwert des Zylinders.

Die Bestimmung des Drehmoments  $M$  erfolgt analog zu obigen Ausführungen über die beschleunigende Kraft  $F_x$  (siehe (23)) bei zeitlich veränderlichem Trägheitmoment  $I = I(t)$  aus dem Drehimpuls  $L = L(t) = I\dot{\varphi}$ . Es gilt

$$M = \frac{dL}{dt} = \frac{d}{dt}(I\dot{\varphi}) = \dot{I}\dot{\varphi} + I\ddot{\varphi} = F_M r. \quad (29)$$

Drückt man aus der letzten Beziehung wiederum  $F_M$  aus, erhält man

$$F_M = \frac{\dot{I}\dot{\varphi} + I\ddot{\varphi}}{r}. \quad (30)$$

Zur Auswertung von  $F_M$  sei auf die diesen Abschnitt einleitenden Überlegungen, insbesondere auf die Beziehungen (13) bis (22), verwiesen.

---

<sup>3</sup>Wie in der Skizze dargestellt wird angenommen, dass der Luftwiderstandskraftvektor parallel zur schiefen Ebene liegt. D.h., dass die Bewegung des Zylinderschwerpunktes in  $y$ -Richtung bei der Berechnung der Luftwiderstandskraft unberücksichtigt bleibt.

Für die Bewegung in  $x$ -Richtung (siehe (24)) gilt somit

$$\begin{aligned}
f(x, \dot{x}, \ddot{x}) &:= m\dot{x} + m\ddot{x} - mg \sin(\alpha) \\
&+ \frac{I\dot{\varphi} + I\ddot{\varphi}}{r} \\
&+ \mu(m\dot{y} + m\ddot{y} + mg \cos(\alpha)) \\
&+ \frac{1}{2}\rho_L c_w A \dot{x}^2 = 0,
\end{aligned} \tag{31}$$

wobei für  $\dot{y} = \dot{r}$  und  $\ddot{y} = \ddot{r}$  (siehe (14) und (15)) gelten, da die  $y$ -Koordinate des Zylinderschwerpunktes stets ident mit der Größe des Radius  $r$  (siehe (13)) des Zylinders ist.

**Bemerkung:** Setzt man (13) bis (22) in (31) ein, so erkennt man, dass man statt eines Systems von Differentialgleichungen nur eine (skalare) Differentialgleichung zu lösen hat. Die Bewegungsgleichung in  $y$ -Richtung (25) wird zur Berechnung der Normalkraft  $F_N$  (siehe (27)) benötigt. Die Funktion  $y = y(t) = r(t)$  wird aus der Lösung der Bewegungsgleichung (31),  $x = x(t)$ , errechnet. Die linke Seite von (31) hängt nur von  $x$  und seinen Ableitungen ab, d.h.  $f = f(x, \dot{x}, \ddot{x})$ .

### 3.2 Lösung der Bewegungsgleichung

Die Bewegungsgleichung (31) ist eine nichtlineare implizite gewöhnliche Differentialgleichung zweiter Ordnung, für dessen Lösung zwei Anfangswerte  $x(t = 0) = \xi_0$  und  $\dot{x}(t = 0) = \dot{\xi}_0$  erforderlich sind. Zur numerischen Lösung mittels finiter Differenzen diskretisiert man  $f(x, \dot{x}, \ddot{x})$  auf einem äquidistanten Gitter mit der Schrittweite  $\Delta t$ . Für die einzelnen Gitterpunkte  $t_i$  im Intervall  $[t_0, t_{end}]$  gilt  $t_i = i \Delta t$  mit  $i = 0, \dots, N$  und  $N = \frac{t_{end}}{\Delta t}$ .

- Für den Startschritt  $i = 1$  bzw.  $t_1 = \Delta t$  gilt

$$f\left(\xi_0, \dot{\xi}_0, \frac{\xi_1 - \xi_0}{\Delta t} - \frac{\dot{\xi}_0}{2}\right) = 0. \tag{32}$$

Die so gewonnene nichtlineare Gleichung wird mittels Newton-Verfahren nach  $\xi_1$  aufgelöst.

- Für  $i \geq 2$  werden die Funktion  $x(t)$  und deren Ableitungen an der Stelle  $t_{i-1}$  wie folgt approximiert<sup>4</sup>

$$x(t_{i-1}) \approx \xi_{i-1}, \tag{33}$$

$$\dot{x}(t_{i-1}) \approx \frac{\xi_i - \xi_{i-2}}{2\Delta t}, \tag{34}$$

$$\ddot{x}(t_{i-1}) \approx \frac{\xi_i - 2\xi_{i-1} + \xi_{i-2}}{\Delta t^2}. \tag{35}$$

Eingesetzt in (31) erhält man

$$f\left(\xi_{i-1}, \frac{\xi_i - \xi_{i-2}}{2\Delta t}, \frac{\xi_i - 2\xi_{i-1} + \xi_{i-2}}{\Delta t^2}\right) = 0. \tag{36}$$

Daraus wird mittels Newton-Verfahren Schritt für Schritt die Folge der  $\xi_i$  berechnet.


---

<sup>4</sup>Beide Differenzenquotienten haben Konsistenzordnung 2.

### 3.3 Beispiel 1


Zur Überprüfung obiger Ausführungen wurde ein Maxima-Programm erstellt. Die Graphen zeigen die Ergebnisse für folgende Parameter:

- Neigungswinkel der Ebene  $\alpha = 40^\circ$
- Startwert für den Radius  $r_0 = 0,3 \text{ m}$
- Koeffizient zur Steuerung der Zunahme des Radius  $C = 0,05$
- Höhe des Zylinders = Breite der Walze  $h = 10 \text{ m}$
- Dichte des Zylinders  $\rho = 500 \frac{\text{kg}}{\text{m}^3}$
- Rollreibungskoeffizient  $\mu = 0,3$
- Dichte der Luft  $\rho_L = 1,29 \frac{\text{kg}}{\text{m}^3}$
- Widerstandsbeiwert des Zylinders  $c_w = 1,0$
- Startwert für den Weg in  $x$ -Richtung  $x(t=0) = \xi_0 = 0 \text{ m}$
- Startwert für die Geschwindigkeit in  $x$ -Richtung  $\dot{x}(t=0) = \dot{\xi}_0 = 0 \frac{\text{m}}{\text{s}}$
- Erdbeschleunigung  $g = 9,81 \frac{\text{m}}{\text{s}^2}$
- Integrationsdauer  $t_{end} = 4,5 \text{ s}$
- (Zeit-)Schrittweite  $\Delta t = 0,05 \text{ s}$




Weg  $x$  in  $m$  aufgetragen über der Zeit  $t$  in  $s$ .







Geschwindigkeit  $\dot{x}$  in  $\frac{m}{s}$  aufgetragen über der Zeit  $t$  in  $s$ .



Beschleunigung  $\ddot{x}$  in  $\frac{m}{s^2}$  aufgetragen über der Zeit  $t$  in  $s$ .



Radius  $r$  ( $=y$ -Koordinate von  $S$ ) des Zylinders in  $m$  aufgetragen über der Zeit  $t$  in  $s$ .



$x$ - $y$ -Plot der Schwerpunktskoordinaten in  $m$ .


**Bemerkung:** Die Graphen verstehen sich als Näherungslösungen (Folge der  $\xi_i$ , bzw. der zugehörigen ersten und zweiten Differenzenquotienten, siehe (34) und (35)).

### 3.4 Beispiel 2


Setzt man unter Beibehaltung sämtlicher anderer Parameter aus Beispiel 1 die Parameter

- Koeffizient zur Steuerung der Zunahme des Radius  $C = 0$ ,
- Rollreibungskoeffizient  $\mu = 0$ ,
- Widerstandsbeiwert des Zylinders  $c_w = 0$ ,


so erhält man als Spezialfall die Lösung für Abschnitt 2.2 mit konstantem Radius  $r$ . Es folgt ein Vergleich von exakter und numerischer Lösung. Man sieht, dass die numerische Lösung die gemäß den theoretischen Überlegungen vorhergesagte gleichmäßig beschleunigte Bewegung liefert.



Weg  $x$  in  $m$  aufgetragen über der Zeit  $t$  in  $s$ .



Geschwindigkeit  $\dot{x}$  in  $\frac{m}{s}$  aufgetragen über der Zeit  $t$  in  $s$ .



Beschleunigung  $\ddot{x}$  in  $\frac{m}{s^2}$  aufgetragen über der Zeit  $t$  in  $s$ .

## A Anhang – Quellcode des Maxima-Programms

Es folgt der Quellcode des Maxima-Programms zur Lösung der Bewegungsgleichungen aus Abschnitt 3.

```
kill(all);
load("mnewton");
ratprint:false$

alpha:40/180*%pi /*Neigungswinkel der Ebene, alle Eingaben in SI-Einheiten*/;
r0:0.3 /*Startwert fuer den Radius*/;
C:0.05 /*Koeffizient zur Steuerung der Zunahme des Radius*/;
h:3 /*Hoehe des Zylinders = Breite der Walze*/;
rho:500 /*Dichte des Zylinders*/;
mu:0.3 /*Reibungskoeffizient*/;
rhoL:1.29 /*Dichte der Luft*/;
cw:1; /*Widerstandsbeiwert des Zylinders*/
xi[0]:0 /*Startwert fuer den Weg in x-Richtung*/;
xidot[0]:0 /*Startwert fuer die Geschwindigkeit in x-Richtung*/;
g:9.81 /*Erdbeschleunigung*/;
tend:4.5 /*Gesamtdauer der (numerischen) Integration*/;
deltat:0.05 /*(Zeit-)Schrittweite*/;

N:round(tend/deltat) /*Anzahl der Schritte*/;

r(x):=r0+C*x /*Annahme: lineare Zunahme des Radius
              in Abhaengigkeit des zurueckgelegten Weges*/;
rdt(xdt):=C*xdt;
rddt(xddt):=C*xddt;

phi(x):=x/r(x);
phidt(x,xdt):=(xdt-rdt(xdt)*phi(x))/r(x);
phiddt(x,xdt,xddt):=(xddt-rddt(xddt)*phi(x)-2*rdt(xdt)*phidt(x,xdt))/r(x);

m(x):=rho*r(x)^2*%pi*h /*Masse des Zylinders*/;
mdt(x,xdt):=rho*%pi*h*2*r(x)*rdt(xdt);
I(x):=1/2*rho*%pi*h*r(x)^4 /*Traegheitsmoment des Zylinders*/;
Idt(x,xdt):=2*rho*%pi*h*r(x)^3*rdt(xdt);

FM(x,xdt,xddt):=(Idt(x,xdt)*phidt(x,xdt)+I(x)*phiddt(x,xdt,xddt))/r(x)
/*Kraft aufgrund der Drehung*/;

FN(x,xdt,xddt):=mdt(x,xdt)*rdt(xdt)+m(x)*rddt(xddt)+m(x)*g*cos(alpha)
/*Normalkraft auf die schiefe Ebene*/;

FR(x,xdt,xddt):=mu*FN(x,xdt,xddt) /*Rollreibunskraft*/;
```

```

FL(x,xdt):=1/2*rhoL*cw*h*2*r(x)*xdt^2 /*Luftwiderstandskraft,
                                         Schattenflaeche A=h*2*r(x) */;

F(x,xdt,xddt):=mdt(x,xdt)*xdt+m(x)*xddt-m(x)*g*sin(alpha)
               +FM(x,xdt,xddt)+FR(x,xdt,xddt)+FL(x,xdt)
/*linke Seite der impliziten Bewegungsgleichung*/;

glg:F(xi[0],xidot[0],((xi[1]-xi[0])/deltat-xidot[0])/(deltat/2))=0
/*Startschritt-Loesung*/;
hilf:mnewton([glg],[xi[1]],[xi[0]]);
xi[1]:rhs(hilf[1][1]);

for i:2 step 1 thru N do block( /*Loesung*/
  glg:F(xi[i-1],(xi[i]-xi[i-2])/(2*deltat),
        (xi[i]-2*xi[i-1]+xi[i-2])/deltat^2)=0,
  hilf:mnewton([glg],[xi[i]],[xi[i-1]]),
  xi[i]:rhs(hilf[1][1])
);

xiliste:makelist([i*deltat,xi[i]],i,0,N) /*Liste der x-Werte*/$

for i:1 step 1 thru N-1 do block( /*Liste des xdt-Werte*/
  xidot[i]:(xi[i+1]-xi[i-1])/(2*deltat)
);
xidotliste:makelist([i*deltat,xidot[i]],i,0,N-1)$

for i:0 step 1 thru N-2 do block( /*Liste der xddt-Werte*/
  xiddot[i]:(xi[i+2]-2*xi[i+1]+xi[i])/deltat^2
);
xiddotliste:makelist([i*deltat,xiddot[i]],i,0,N-2)$

for i:0 thru N do y[i]:r0+C*xi[i] /*Berechnung der y-Koordinate
                                   des Schwerpunkts*/;

xiyliste:makelist([xi[i],y[i]],i,0,N) /*Liste fuer den x-y-Plot*/$

yliste:makelist([i*deltat,y[i]],i,0,N) /*Liste der y-Werte
                                         (=momentaner Radius des Zylinders)*/$

wxplot2d([[discrete,xiliste]], [legend,""], [xlabel,"t"], [ylabel,"x"])$
plot2d([[discrete,xiliste]], [legend,""], [xlabel,"t"], [ylabel,"x"],
        [gnuplot_term,ps], [gnuplot_out_file,"x-t.eps"])$

wxplot2d([[discrete,xidotliste]], [legend,""], [xlabel,"t"], [ylabel,"dx/dt"])$
plot2d([[discrete,xidotliste]], [legend,""], [xlabel,"t"],
        [ylabel,"dx/dt"], [gnuplot_term,ps], [gnuplot_out_file,"xdt-t.eps"]);

```

```

wxplot2d([[discrete,xiddotliste]], [legend,""], [xlabel,"t"],
         [ylabel,"d^2x/dt^2"])$
plot2d([[discrete,xiddotliste]], [legend,""], [xlabel,"t"],
       [ylabel,"d^2x/dt^2"],
       [gnuplot_term,ps], [gnuplot_out_file,"xddt-t.eps"]);

wxplot2d([[discrete,yliste]], [legend,""], [xlabel,"t"], [ylabel,"r"])$
plot2d([[discrete,yliste]], [legend,""], [xlabel,"t"], [ylabel,"r"],
       [gnuplot_term,ps], [gnuplot_out_file,"r-t.eps"]);

wxplot2d([[discrete,xiyliste]], [legend,""])$
plot2d([[discrete,xiyliste]], [legend,""],
       [gnuplot_term,ps], [gnuplot_out_file,"x-y.eps"]);

```